

Title	リカーシブによる自己相似三次元図形の試み
Author(s)	森, 久紘
Citation	デザイン理論. 1992, 31, p. 88-89
Version Type	VoR
URL	https://doi.org/10.18910/53265
rights	
Note	

Osaka University Knowledge Archive : OUKA

<https://ir.library.osaka-u.ac.jp/>

Osaka University

リカーシブによる自己相似三次元図形の試み

森 久紘

*はじめに

デザインや造形などのドローイングにコンピュータを利用することにより、今までにないデザイン制作や、デザインプロセスの改革につながる。そこでまず第一にUNIX X-Window上のグラフィックスの関数群およびパソコンにおいてはBASIC言語によるサブルーチン群を作成することにより、比較的簡単にプログラムが可能となることを目指した。第二にこれらグラフィックス関数群およびリカーシブを使ってまず二次元の図形の作成プログラムを試みた。短いプログラムにより、複雑な図形を作成することができる。そして第三に上記と同様の手法であるが、三次元図形のプログラムを作成し表示させた。

*基本的な関数群について

これら関数群は三次元グラフィックスの関数群であり、それらの中にはタートル・グラフィックスといわれる機能も含まれている。これはタートルという一種のカーソルをロボットのように移動させ、移動軌跡を線状に画面上に描かせるものである。タートルの進む方向（現在の向きに対して何度向きを変えるか）、進ませる距離を指示することによって移動させるものである。たとえば一辺が100である正方形を描かせる場合、通常絶対座標はX, Yの座標値(0, 0)から描きはじめるとすると、(0, 0), (100, 0), (100, 100), (0, 100), (0, 0)と順次結んでゆく。タートル・グラフィックでは、タートルを100

前進させて右へ90度回転させ向きを変えるという動作を4回繰り返せばよい。また同様にある座標値からタートルの向きを変えないで、前へ100, 右へ100, 後へ100, 左へ100と移動させても正方形を描くことができる。絶対座標値による図形はただの一つしかない。これに対して後者二つの方法はいずれも相対座標値による図形である。したがって自由な座標値で自由な向きに図形を描くことができる。これらの機能を三次元に拡張したものを関数群としてもっている。つまりZ座標値の方向も指示することによって、三次元図形（立体図形）を作図することができるのである。

*二次元自己相似図形

まず簡単な二次元図形を考えてみよう。

a. 簡単な曲線の描画

タートル・グラフィックスによって簡単な曲線を簡単に描くことができる。たとえば螺旋の場合は初期値に順次、加算または減算あるいは乗除させる場合、さらに平方根の値による変化などによって描かせる場合が考えられる。変化させるものは進む距離あるいは向きの角度である。

b. 自己相似図形の描画

先記関数群を使用すると、図1のような図形もプログラム1のように短いステップで作図できるだけでなく、図形とプログラムの対応関係が比較的わかりやすいので感覚的にプログラミングできる。970行から1010行までは初期設定とサブルーチンの*SSを呼出しており、1040行、1080行、

1100行はそれぞれ三分の一の円弧を描いている。1070行、1090行、1120行はリカーシブ・コールをおこなっており、小さな子供の円弧を描く。1070行の時のみ180度回転させてから描いている。




図1

プログラム1

```

970 ' EN2 1988,1,16 FBASIC 16B
980 MERGE "b:TUR",R
1000 *MAIN
1005 GOSUB*SETP:GOSUB*EYE:COL=7
1010 GOSUB*HOME:MV=20:GOSUB*SS:END
1020 *SS
1030 IF MV<.5 THEN RETURN
1040 GOSUB*EN
1045 TN=180:GOSUB*TN
1070 MV=MV/2.5:GOSUB*SS:MV=MV*2.5
1075 TN=-180:GOSUB*TN
1080 GOSUB*EN
1090 MV=MV/2.5:GOSUB*SS:MV=MV*2.5
1100 GOSUB*EN
1120 MV=MV/2.5:GOSUB*SS:MV=MV*2.5
1130 RETURN
2000 *EN
2010 FOR I=1 TO 12
2020 TN=10:GOSUB*TN:GOSUB*MV
2030 NEXT I
2040 RETURN

```

***三次元自己相似図形の作成**

プログラム2は図2のような図形をワイヤフレームで描くプログラムの主要部分である。1000行から1017行は初期設定とサブルーチン*SSの呼び出しで、変数MV

は立方体の一辺の長さであり、Nは何回呼び出すかである。サブルーチン*RIPOは立方体を描くサブルーチンであり、*SPはスピン回転、*UPは手前あるいは向こうへの回転、*TNは右または左回転である。基本的には三次元の場合と同じ手法である。




図2

プログラム2

```

970 'Trs_0
980 MERGE "MOKUBA",R
1000 *MAIN
1005 GOSUB*SETP:GOSUB*EYE
1010 GOSUB*HOME:COL=-1:MV=-150:GOSUB*MV
1015 UP=20:GOSUB*UP:SP=30:GOSUB*SP
1017 MV=100:N=5:BAI=1.6:COL=6:GOSUB*SS:END
1020 *SS
1030 IF N=0 THEN RETURN
1035 N=N-1
1040 GOSUB*RIPO:GOSUB*MV
1042 SP=180:GOSUB*SP
1045 MV=1/BAI*MV:GOSUB*SS:MV=BAI*MV
1047 SP=-180:GOSUB*SP
1050 UP=-90:GOSUB*UP:GOSUB*MV
1055 UP=90:GOSUB*UP:SP=90:GOSUB*SP
1060 MV=1/BAI*MV:GOSUB*SS:MV=BAI*MV
1065 SP=-90:GOSUB*SP:UP=-90:GOSUB*UP
1070 TN=-90:GOSUB*TN:GOSUB*MV
1083 UP=90:GOSUB*UP:SP=90:GOSUB*SP
1085 MV=1/BAI*MV:GOSUB*SS:MV=BAI*MV
1087 SP=-90:GOSUB*SP:UP=-90:GOSUB*UP
1090 TN=-90:GOSUB*TN:GOSUB*MV
1095 UP=90:GOSUB*UP:SP=90:GOSUB*SP
1100 MV=1/BAI*MV:GOSUB*SS:MV=BAI*MV
1110 SP=-90:GOSUB*SP:UP=-90:GOSUB*UP
1118 TN=-90:GOSUB*TN:GOSUB*MV
1119 TN=-90:GOSUB*TN:UP=90:GOSUB*UP
1120 MV=-MV:GOSUB*MV:MV=-MV
1125 N=N+1
1130 RETURN

```

***最後に**

以上、要点の説明をおわる。

もり・ひさひろ 滋賀女子短期大学
1992. 3. 28 CAD分科会